Public Class MyNode

    Private mName As String

    Private mSection As String

    Private mBirthday As DateTime

    Private mNext As MyNode

    'Constructor

    Public Sub New()

        mName = String.Empty

        mSection = String.Empty

        mBirthday = CDate("Jan 1 1900")

        mNext = Nothing

    End Sub

    Public Sub New(ByVal Name As String)

        mName = Name.Trim

        mSection = String.Empty

        mBirthday = CDate("Jan 1 1900")

        mNext = Nothing

    End Sub

    'Properties

    Public Property Name() As String

        Get

            Return mName

        End Get

        Set(ByVal value As String)

            mName = value

        End Set

    End Property

    Public Property Section() As String

        Get

            Return mSection

        End Get

        Set(ByVal value As String)

            mSection = value

        End Set

    End Property

    Public Property Birthday() As DateTime

        Get

            Return mBirthday

        End Get

        Set(ByVal value As DateTime)

            mBirthday = value

        End Set

    End Property

    Public Property [Next]() As MyNode

        Get

            Return mNext

        End Get

        Set(ByVal value As MyNode)

            mNext = value

        End Set

    End Property

End Class

Public Class MyLinkedList

    Private mHead As MyNode

    Public Sub New()

        mHead = Nothing

    End Sub

    Public Property Head() As MyNode

        Get

            Return mHead

        End Get

        Set(ByVal value As MyNode)

            mHead = value

        End Set

    End Property

    Public Sub AddFirst(ByVal Node As MyNode)

        Node.Next = mHead

        mHead = Node

    End Sub

    Public Sub AddLast(ByVal Node As MyNode)

        'Adds to the end of the linked list

        Dim Temp As MyNode

        'Start at the top of the linked list

        Temp = mHead

        If IsNothing(mHead) Then

            'The linked list is empty, use AddFirst instead

            AddFirst(Node)

        Else

            'Repeat until the last node is found

            'Because the last node's Next is nothing

            Do Until IsNothing(Temp.Next)

                Temp = Temp.Next

            Loop

            'Assign Node after Temp (the previous last node)

            Temp.Next = Node

        End If

    End Sub

    Public Sub RemoveAfter(ByVal ThisNode As MyNode)

        'Removes the next node after ThisNode

        'Make sure that this is not the last node

        If Not IsNothing(ThisNode.Next) Then

            Dim Temp As MyNode

            Temp = ThisNode.Next.Next

            ThisNode.Next.Next = Nothing

            ThisNode.Next = Temp

        End If

    End Sub

    Public Sub RemoveAllAfter(ByVal ThisNode As MyNode)

        'Removes all nodes after ThisNode

        'Not the last node?

        If Not IsNothing(ThisNode.Next) Then

            Dim Temp As MyNode, Temp2 As MyNode

            Temp = ThisNode.Next

            'Split the linked list after ThisNode

            ThisNode.Next = Nothing

            'Obliterate the rest of the linked list

            Do Until IsNothing(Temp.Next)

                Temp2 = Temp.Next

                Temp.Next = Nothing

                Temp = Temp2

            Loop

        End If

    End Sub

    Public Function Length() As Integer

        'Returns the total length of the linked list

        'Empty list?

        Dim Temp As MyNode

        Dim Counter As Integer

        Counter = 0

        Temp = mHead

        Do Until IsNothing(Temp)

            Counter += 1

            Temp = Temp.Next

        Loop

        Return Counter

    End Function

    Public Sub InsertAfter(ByVal ThisNode As MyNode, ByVal NewNode As MyNode)

        'Inserts after ThisNode a NewNode in the linked list

        NewNode.Next = ThisNode.Next

        ThisNode.Next = NewNode

    End Sub

    Public Function Item(ByVal Index As Integer) As MyNode

        'Returns the node at the specified index

        'If index is illegal (< 0, or exceeds the length of the linked list

        'return Nothing

        'Is the linked list empty?

        If IsNothing(mHead) Then

            Return Nothing

        Else

            Dim Counter As Integer

            Dim Temp As MyNode

            Temp = mHead

            'Skip over index 0, because index 0 is Head

            For Counter = 1 To Index

                'Have we stepped outside the linked list?

                If IsNothing(Temp) Then

                    Return Nothing

                Else

                    Temp = Temp.Next

                End If

            Next

            Return Temp

        End If

    End Function

    Public Function IndexOf(ByVal Name As String) As Integer

        'Returns the index of the first node with Name in it

        'Returns -1 if not found (or if linked list is empty)

        Return IndexOf(Name, 0)

    End Function

    Public Function IndexOf(ByVal Name As String, ByVal StartingIndex As Integer) As Integer

        'Returns the index of the first node with Name in it starting

        'from StartingIndex

        'Returns -1 if not found after the StartingIndex

        'Is the linked list worth checking out?

        If IsNothing(mHead) Then

            Return -1

        End If

        Dim Temp As MyNode

        Dim Counter As Integer = StartingIndex

        Temp = Item(StartingIndex)

        'We can't test the Name property here, just in case Temp turns out

        'to be Nothing (and get a null reference exception!)

        Do Until IsNothing(Temp)

            'Use case-insensitive compare

            If String.Compare(Name, Temp.Name, True) = 0 Then

                Return Counter

            End If

            Counter += 1

            Temp = Temp.Next

        Loop

        'If we're here, we didn't find anything

        Return -1

    End Function

End Class

